Int. J. Multiphase Flow Vol. 9, No. 4, pp. 437–446, 1983
Printed in Great Britain.

0301-9322/83 $3.00 + .00
Pergamon/Elsevier

AN ALGORITHM FOR HYDRODYNAMICS OF TURBULENT UPWARD FLOWING DILUTE GAS-SOLIDS SUSPENSIONS

TÜLAY A. ÖZBELGE
Department of Chemical Engineering, Middle East Technical University, Ankara, Turkey

(Received 24 August 1982; in revised form 30 November 1982)

Abstract—In this paper, the hydrodynamics of dilute gas-solids suspensions is theoretically analyzed. A computer program and its algorithm, which can be applied to any dilute upward flowing suspension in a vertical pipe, are presented. The relative velocity between the two phases, the voidage, the external force acting on the solids due to particle-particle and particle-wall interactions can be calculated for each different solids loading ratio (SLR). The theoretical results are in satisfactory agreement with experimental data.

1. INTRODUCTION

The study of the flow characteristics of gas-solids suspensions is of great importance in many fields of chemical and mechanical engineering. It is encountered in fluidized beds, pneumatic conveying, nuclear reactor cooling, and dust collection.

The previous experimental work by (Metha *et al.* 1957; Vogt & White 1948; Hariu & Molstad 1949; Depew 1960) have been devoted to the determination of the average properties of such two-phase systems. They measured the mass fluxes of solids and gas phases, the pressure drops at each different SLR in the test section, and the weight of solids trapped between the two quick-closing valves in a section of pipe to determine the dispersed solids density, ρ_{ds}. The correlations obtained from the experimental data were partially successful for only limited ranges of parameters.

The later studies required advanced experimental techniques (Boothroyd 1971; Riethmuller 1973; Oki *et al.* 1977) to measure the local properties of two-phase suspension flows, thus in turn, to explain their behaviour. These methods are expensive and they are still in the developmental stages, Riethmuller (1981).

Therefore, the purpose of this work is to present a theoretical analysis for the hydrodynamics of dilute upward flowing gas-solids suspensions in vertical pipes.

2. MATHEMATICAL FORMULATION

The macroscopic momentum balances along the axial direction are written for both of the phases. A simplifying assumption has to be used here to make the problem mathematically tractable, which is that all of the flow parameters are uniform in the radial direction. This is justifiable for a dilute gas-solids suspension flowing in the fully developed region of turbulent flow in a vertical pipe. For a pipe section of differential length Δx, these equations are:

(a) for fluid phase

$$\underbrace{(\pi R_T^2\rho_f u_f)(u_f)|_{x=0}}_{\text{Momentum influx}} \;\underbrace{-(\pi R_T^2\rho_f u_f)(u_f)|_{x=\Delta x}}_{\text{Momentum efflux}} \;\underbrace{-g\pi R_T^2\rho_f\Delta x}_{\text{Gravitational force}}$$

$$\underbrace{-(2\pi R_T\Delta x)\left(\frac{1}{2}\rho_f u_f^2\right)f_w}_{\text{Friction force between fluid and wall}} \underbrace{-n_p(\pi R_T^2\Delta x)A_p\left(\frac{1}{2}\rho_G u_r^2\right)C_D}_{\text{Drag force between solid particles and fluid}}$$

$$\underbrace{+\pi R_T^2 P|_{x=0}-\pi R_T^2 P|_{x=\Delta x}}_{\text{Pressure forces}} \underbrace{+F_{e,f}\Delta x}_{\text{Other external forces on fluid}} = 0 \qquad [1]$$

where R_T = pipe radius; ρ_f = density of fluid phase; u_f = velocity of fluid in the x-direction; g = acceleration of gravity; f_w = friction factor between fluid and pipe wall; n_p = number of particles per unit volume of system; A_p = projected area of a spherical solid particle with a radius R_p, πR_p^2; ρ_G = density of gas; u_r = relative velocity between the phases, also called drag or slip velocity; P = pressure; $F_{e,f}$ = other external forces on the fluid phase; C_D = drag coefficient for a single particle in an assemblage of particles of voidage α.

If [1] is divided by Δx and the limit is taken as Δx approaches zero, the following equation is obtained,

$$\pi R_T^2 \rho_f u_f \frac{du_f}{dx} + g\pi R_T^2 \rho_f + \pi R_T \rho_f u_f^2 f_w + \pi R_T^2 \frac{n_p A_p}{2} \rho_G u_r^2 C_D + \pi R_T^2 \frac{dP}{dx} = 0. \qquad [2]$$

$F_{e,f}$ in [1] is equal to zero, since there are not any other forces acting on the fluid phase than those taken into consideration. Therefore, it does not appear in [2].

Similarly, (b) for solids phase

$$\pi R_T^2 \rho_{ds} u_{ds} \frac{du_{ds}}{dx} + g\pi R_T^2 \rho_{ds} - \pi R_T^2 \frac{n_p A_p}{2} \rho_G u_r^2 C_D + F_{e,s} = 0 \qquad [3]$$

where ρ_{ds} = dispersed solids density; u_{ds} = solids phase velocity; $F_{e,s}$ = external forces on the solids which are electrostatic forces and the particle-wall, particle-particle interaction forces.

Equations [2] and [3] can be solved by means of other independent equations which are:

The equation of continuity is,

$$\frac{\rho_f}{\rho_G} + \frac{\rho_{ds}}{\rho_p} = 1 \qquad [4]$$

where ρ_p is the material density of solids particles. The voidage of the system is found from the consideration that each phase is a continuum;

$$\alpha = \frac{\rho_f}{\rho_G} \qquad [5]$$

$$1 - \alpha = \frac{\rho_{ds}}{\rho_p}. \qquad [6]$$

The mass fluxes of gas and solids phases are respectively,

$$G_G = \rho_G u_G \qquad [7]$$

$$G_{ds} = \rho_{ds} u_{ds}. \qquad [8]$$

In [7], u_G is the superficial gas velocity which is equal to

$$u_G = \alpha u_f \qquad [9]$$

The number of particles per unit volume of the system is,

$$n_p = (1 - \alpha)/V_p \qquad [10]$$

where V_p is the volume of a spherical solid particle. Throughout this work,spherical particles have been considered; a shape factor has not been taken into account.

For relatively dense gas-solids mixtures the following relationship given by Wen & Galli (1971) is used:

$$C_D = C_{D_t}\alpha^{-4.7} \qquad [11]$$

where C_{D_t} is the drag coefficient for a single particle in an unbounded air stream and C_D is the drag coefficient for a single particle in an assemblage of particles of voidage α. For lightly-loaded systems, C_D approaches to C_{D_t} which is given by standard charts for drag coefficients and it can be expressed as a function of particle Reynolds number,

$$C_{D_t} = \text{Func}\left(\frac{2R_p\rho_G u_r}{\mu_G}\right) \qquad [12]$$

where μ_G is the viscosity of gas.

The other assumptions used in the solution of the problem are: (1) gas phase obeys the ideal gas law,

$$\rho_G = \frac{PM}{RT} \qquad [13]$$

where M = molecular weight of gas; R = gas constant; T = absolute temperature; P = pressure in the system; (2) pressure of the system varies linearly with axial distance, x; (3) although the properties ρ_f, ρ_{ds}, u_f and u_{ds} are functions of x, they do not change much with x in the fully developed flow region of the pipe; therefore their averaging in the axial direction can be justified; (4) since ρ_p is much greater than ρ_G buoyancy effects are neglected; (5) assuming that the fluid mechanics of the gas phase is unaffected by the presence of solids particles in very dilute gas-solids suspensions, the following relationship given by Bird *et al.* (1960) for the gas-wall friction factor, f_w, of a single phase turbulent flow can be used here,

$$\frac{1}{\sqrt{(f_w)}} = 4\log_{10}(\text{Re}\sqrt{(f_w)}) - 0.40 \qquad [14]$$

where Re is the Reynolds number; (6) void fraction, α, is a constant for a set of experimental conditions in the fully developed flow region; justification for this assumption is based on the experimental results of Kramer and Depew (1972). Multiplication of [2] with $(1/\pi R_T^2\rho_f)\,dx$ and substitution of [5], [10], [13] in [2] yields [15] in the integration form,

$$\int_{u_{f_1}}^{u_{f_2}}(u_f\,du_f) + \int_0^L g\,dx + \int_0^L \frac{u_f^2 f_w}{R_T}\,dx + \int_0^L (1-\alpha)\frac{u_r^2 C_D A_p}{2\alpha V_p}\,dx + \int_{P_1}^{P_2}\left(\frac{RT}{\alpha M}\frac{dP}{P}\right) = 0. \qquad [15]$$

For a spherical particle,

$$\frac{A_p}{V_p} = \frac{3}{4R_p}. \qquad [16]$$

Equation [15] is integrated analytically using the average values of the system's parameters between the pressure taps "1" and "2", according to the assumption (3); then it is combined with [16] to obtain

$$\frac{1}{2}(u_{f_2}^2 - u_{f_1}^2) + gL + \frac{\langle u_f\rangle^2 f_w L}{R_T} + \frac{3}{8}\frac{(1-\alpha)}{\alpha}C_D\langle u_r\rangle^2\frac{L}{R_p} + \frac{R\langle T\rangle}{\alpha M}\ln\frac{P_2}{P_1} = 0 \qquad [17]$$

where $\langle\ \rangle$ sign represents the average in the axial direction between the pressure taps "1" and "2" and L is the length of the vertical test section between the pressure taps. In [17], the average slip velocity, $\langle u_r \rangle$, is expressed as

$$\langle u_r \rangle = \langle u_f \rangle - \langle u_{ds} \rangle \qquad [18]$$

Equations [5], [7] and [13] are combined to obtain the velocity of fluid phase between particles,

$$u_f = \frac{G_G RT}{\alpha MP} \qquad [19]$$

Substitution of [19] in the first term of [17] yields

$$\frac{1}{2}\left(\frac{G_G R}{\alpha M}\right)^2 \left\{\left(\frac{T_2}{P_2}\right)^2 - \left(\frac{T_1}{P_1}\right)^2\right\} + gL + \frac{\langle u_f \rangle^2 f_w L}{R_T}$$
$$+ \frac{3}{8}\frac{(1-\alpha)}{\alpha} C_D(\langle u_f \rangle - \langle u_{ds} \rangle)^2 \frac{L}{R_p} + \frac{R\langle T \rangle}{\alpha M} \ln\frac{P_2}{P_1} = 0 \qquad [20]$$

After the drag or slip velocity, $\langle u_r \rangle = \langle u_f \rangle - \langle u_{ds} \rangle$, is calculated from [20], an expression for the external force acting on a single solid particle can be derived as follows:

Multiplication of [3] with $((1/\pi R_T^2)(\rho_p/\rho_{ds})V_p)$ and substitution of [6], [10], in [3] yields a force balance for a single particle,

$$\rho_p V_p u_{ds} \frac{du_{ds}}{dx} = \frac{1}{2} A_p \rho_G \langle u_r \rangle^2 C_D - g\rho_p V_p - f_{e,s} \qquad [21]$$

where $f_{e,s}$ is the external force per solid particle which is defined as,

$$f_{e,s} = F_{e,s} \frac{\rho_p}{\pi R_T^2 \rho_{ds}} V_p \qquad [22]$$

In [22], $F_{e,s}$ is the total external force acting on the whole solids phase per unit length of test section.

The assumption (6) implies that the left hand side of [21] is zero; because from [6] and [8] u_{ds} is equal to,

$$u_{ds} = \frac{G_{ds}}{\rho_p(1-\alpha)} \qquad [23]$$

Since $(d\alpha/dx)$ is assumed to be zero, then (du_{ds}/dx) is zero also from [23]. Therefore from [21], $f_{e,s}$ is obtained,

$$f_{e,s} = \frac{1}{2} A_p \rho_G \langle u_r \rangle^2 C_D - g\rho_p V_p. \qquad [24]$$

The algorithm for the calculations of the drag velocity, voidage, solids phase velocity and the external force is presented in the following section.

3. ALGORITHM AND THE COMPUTER PROGRAM

In order to carry out the calculations with the present computer program it is necessary to put in the following information: pipe diameter, particle size and density, length of test section, inlet pressure and temperature of gas into test section, viscosity and molecular weight of gas used in conveying solids particles, outlet temperature of suspension from test section, total pressure drop caused by suspension at each SLR through test section and superficial gas velocity.

An iterative procedure is used in calculating the drag velocity. Terminal velocity of a solid particle is used as an initial guess for drag velocity; by means of that the necessary parameters are calculated from the equations given in the present formulation to be used in [20] to get a new value for the drag velocity which is compared with the value of the previous iteration step. This procedure is repeated until the required convergence between the two successive iterations is obtained. Then the program returns $\langle u_f \rangle$, α, $\langle u_{ds} \rangle$, ρ_{ds}, $\langle u_r \rangle$, $f_{e,s}$ for each set of superficial gas velocity and the value of SLR which is the ratio of the mass flux of solids to the mass flux of gas

$$\mathrm{SLR} = \frac{G_{ds}}{G_G}. \qquad [25]$$

In the light of these computational considerations, a computer program, whose algorithm is given in figure 1, has been written. The nomenclature of the algorithm and the computer program is presented at the beginning of the program.

The Fortran Program has been written in British units; then the necessary conversions to SI units have been provided just before the print out statements.

3.1 *Fortran program*

```
C     ************************************************************
C     PROGRAM CALCULATES DRAG VELOCITY, VOIDAGE AND EXTERNAL FORCE FROM
C     EXPERIMENTAL DATA OF HARIU AND MOLSTAD (1949)
C     ************************************************************
C
C     NOMENCLATURE OF THE PROGRAM
C
C     RT=PIPE RADIUS
C     RP=RADIUS OF SOLID PARTICLE
C     DT=DIAMETER OF PIPE
C     DP=DIAMETER OF PARTICLE
C     PL=LENGTH OF TEST SECTION IN VERTICAL FULLY DEVELOPED SUSPENSION FLOW
C     TO=INLET TEMPERATURE OF GAS
C     PO=INLET PRESSURE
C     TMM=OUTLET TEMPERATURE FROM TEST SECTION
C     TM=MEAN GAS TEMPERATURE IN TEST SECTION
C     WM=MOLECULAR WEIGHT OF GAS
C     DELP=TOTAL PRESSURE DROP DUE TO SUSPENSION FLOW THROUGH TEST SECTION
C     PX=OUTLET PRESSURE FROM TEST SECTION
C     GO=GRAVITATIONAL ACCELERATION
C     ICHECK=IT CHECKS WHETHER THE ASSIGNED NUMBER OF ITERATIONS ARE ACCOMPLISHED
C     IMAX=MAXIMUM NUMBER OF ITERATIONS TO OBTAIN THE CONVERGENCE OF DRAG VELOCITY
C     R=GAS CONSTANT
C     VISC=VISCOSITY OF GAS
C     REF=REYNOLDS NUMBER OF FLUID
C     UDRAG=DRAG VELOCITY (RELATIVE VELOCITY)
C     UDRAGO=VALUE OF DRAG VELOCITY IN PREVIOUS ITERATION STEP
C     FW=FRICTION FACTOR BETWEEN GAS AND PIPE WALL
C     WF=MASS FLUX OF GAS
C     WS=MASS FLUX OF SOLIDS
C     WR=LOADING RATIO (SLR)
C     UFAVE=AVERAGE VELOCITY OF FLUID PHASE
C     USAVE=AVERAGE VELOCITY OF SOLIDS PHASE
C     EPS=VOIDAGE OF GAS
C     RES=PARTICLE REYNOLDS NUMBER
C     FS=DRAG COEFFICIENT BETWEEN GAS AND SOLIDS PARTICLE
C     RHOG=DENSITY OF GAS
C     RHOP=MATERIAL DENSITY OF SOLID
C     FES=EXTERNAL FORCE ACTING ON A SINGLE PARTICLE
C     DAMP=DAMPING FACTOR USED TO PROVIDE CONVERGENCE
C     RHOS=DENSITY OF DISPERSED SOLIDS PHASE
C     RHOF=DENSITY OF FLUID PHASE
C     PM=AVERAGE GAS PRESSURE IN TEST SECTION
C     UGAS=SUPERFICIAL GAS VELOCITY
```

```
C
C     CONSTANTS
C
      GO=32.174*3600.*3600.
      R=1545.33
      WM=29.
C
C     INPUT DATA ARE READ
C
      READ(5,10)RHOP,DP,DT,PL
      READ(5,11)IMAX
    2 READ(5,13)WR,TMM,TO,PO,DELP,UGAS,VISC
      IF(WR.EQ.0.)GO TO 99
      READ(5,17)DAMP
      TO=TO+460.
      TMM=TMM+460.
      TM=(TMM+TO)/2.
      PX=PO-DELP
      PM=PO-(DELP/2.)
      RHOG=PM*WM/R/TM
      VISCO=VISC*242.
      WF=RHOG*UGAS*3600.
      WS=WR*WF
C     ITERATION FOR DRAG VELOCITY STARTS
C     WITH CALCULATION OF TERMINAL VELOCITY
C     AS AN INITIAL GUESS
C
      UTERM=2.*RP*RP*(RHOP-RHOG)*GO/VISCO/9.
      UDRAG=UTERM
      ICHECK=0
    4 RES=UDRAG*2.*RP*RHOG/VISCO
      ICHECK=ICHECK+1
C     STOKE'S REGION
C
      IF(RES.LT.2.) FS=24./RES
C
C     INTERMEDIATE REGION
C
      IF(RES.GE.2..AND.RES.LT.500.) FS=18.5/(RES**0.6)
C
C     NEWTON'S REGION
C
      IF(RES.GE.500.) FS=0.44
      BQ=-UDRAG-WF/RHOG-WS/RHOP
      CQ=UDRAG*WF/RHOG
      UFAVE=-0.5*(BQ-SQRT(BQ*BQ-4.*CQ))
      EPS=WF/UFAVE/RHOG
      RHOF=WF/UFAVE
      REF=2.*RT*UFAVE*RHOF/VISCO
C     CALCULATION OF GAS-WALL FRICTION FACTOR BY NEWTON-RAPHSON ITERATIVE METHOD
      FW=0.007
  100 IF(FW.LE.0.) FW=0.00001
      F=4.*ALOG10(REF*SQRT(FW))-0.4-1.0/SQRT(FW)
      DF=2./FW+0.5*FW**(-1.5)
      FWW=FW-F/DF
      IF (ABS(FWW-FW).LE.1.0 E-6*FWW) GO TO 200
      FW=FWW
      GO TO 100
  200 UDRAGO=UDRAG
      ARG1=GO*R*TM/EPS/WM*ALOG(PO/PX)
      ARG2=(R*WF/EPS/WM)**2/2.*(TO*TO/PO/PO-TMM*TMM/PX/PX)
      ARG3=FW*UFAVE*UFAVE*PL/RT
      ARG4=GO*PL
C     DRAG VELOCITY IS CALCULATED WITH EQUATION OBTAINED FROM MATHEMATICAL
C     DERIVATION
C
      UDRAG=SQRT((8./3.*EPS*RP/(1.-EPS)/FS/PL)*(ARG1+ARG2-ARG3-ARG4))
   50 ABC=UDRAGO-UDRAG
      IF(ICHECK.GT.IMAX) GO TO 300
      IF(ABS(ABC).LE.0001*UDRAG) GO TO 400
      UDRAG=UDRAGO-DAMP*ABC
      GO TO 4
  300 WRITE(6,350)
  350 FORMAT(5X,'*** THIS RUN DID NOT CONVERGE')
C
C     CALCULATION OF EXTERNAL FORCE ON A PARTICLE
C
  400 FES=(4.4482E5/GO*3.1416*RP*RP)*((0.5*RHOG*UDRAG*UDRAG*FS)-(GO*RHOP
     1*4./3.*RP))
C
C     CALCULATION OF SOLIDS PHASE VELOCITY AFTER DRAG VELOCITY
C     HAS BEEN OBTAINED BY ITERATION
C
      USAVE=UFAVE-UDRAG
C
C     CALCULATION OF SOLIDS PHASE DENSITY AFTER SOLIDS PHASE
C     VELOCITY HAS BEEN CALCULATED
```

```
C
      RHOS=WS/USAVE
      RR=RHOS/RHOF
      UDRAG=UDRAG/3600.
      UFAVE=UFAVE/3600.
      USAVE=USAVE/3600.
      UTERM=UTERM/3600.
C
C     CALCULATED VALUES ARE CONVERTED TO (SI) UNITS
C
      FEY=FES/100000.
      VISCY=VISCO/242.
      PY=PO*6894.73/144.
      TY=TM/1.8
      UDRAG=UDRAG*0.3048
      UFAVE=UFAVE*0.3048
      USAVE=USAVE*0.3048
      UTERM=UTERM*0.3048
      RHOS=RHOS*16.018
      RHOF=RHOF*16.018
      RHOG=RHOG*16.018
      RHOY=RHOP*16.018
      PLY=PL*0.3048
      DTY=DT*0.3048
      DPY=DP*0.3048
      WRITE(6,30)
      WRITE(6,31) RHOY,DPY,DTY,PLY
      WRITE(6,42) PY,TY,VISCY
      WRITE(6,12)
      WRITE(6,14) WR,UFAVE,USAVE,RHOG,RHOS,UDRAG,EPS,REF,FW,RES,FS,RR,FE
     *Y
      WRITE(6,15) ICHECK
      WRITE(6,16) UTERM
      GO TO 2
   10 FORMAT(F7.1,F9.5,F7.3,F6.2)
   11 FORMAT(I4)
   12 FORMAT(/6X,'WR',4X,'UFAVE',4X,'USAVE',4X,'RHOG',4X,'RHOS',4X,'UDRA
     1G',6X,'EPS',6X,'REF',6X,'FW',4X,'RES',4X,'FS',4X,'RR',5X,'FES')
   13 FORMAT(6F 10.0,E14.0)
   14 FORMAT(2X,F8.3,F7.1,F8.1,F10.4,F8.3,F9.3,F9.5,F9.0,F8.4,F8.2,F5.2,
     1F8.2,E12.5)
   15 FORMAT(5X,'***NUMBER OF ITERATIONS=',I4)
   16 FORMAT (E11.4)
   17 FORMAT(F10.0)
   30 FORMAT(1H0)
   31 FORMAT(7X,'RHOP',7X,'DP',8X,'DT',6X,'PL'/(4E11.4))
   42 FORMAT(9H       PO=,E11.4,10H        TM=,F4.0,10H    VISCO=,F8.6)
   99 STOP
      END
```

3.2 *Algorithm*



Figure 1. Algorithm of the computer program.

4. RESULTS AND CONCLUSIONS

Since the behaviour of turbulently flowing gas-solids suspensions is very complex, therefore the present theoretical model has been derived with some simplifying assumptions which are reasonable within certain limits. The most critical one of these is the acception of the uniformly distributed dispersed solids phase in the radial direction and in the fully developed flow region. The lower the SLR and the higher the gas phase velocity, the easier is to justify the above assumption. The other important factors which govern the flow are the pipe size, the particle size, the particle density, and the gas density.

As it has been stated before, the model applies to turbulently upward flowing dilute gas-solids suspensions in vertical pipes. Before using the present theoretical model in practical applications, one has to be sure about the reliability of the test section pressure drop data at different values of SLR. The total pressure drop through the test section includes the pressure drops due to gravity, due to gas-wall friction and due to the friction caused by the presence of solid particles. These individual pressure drops are difficult to measure separately and correctly. Therefore, the requirement of the total pressure drop data here makes the problem easier to handle.

The calculated results and the experimental data of Hariu & Molstad (1949) are given in table 1. The present model has been applied to calculate the parameters for two-phase flow system of the experimental study by Depew (1960); then using these results, Nusselt numbers have been calculated by solving the coupled heat and momentum transfer equations numerically for the same system by Özbelge & Somer (1982). The numerical results checked the experimental heat transfer results of Depew (1960) very closely. This shows the usefulness and the accuracy of the present model.

The model checks with the experimental data within almost $\mp 10\%$ error margin for the following intervals of the parameters: voidage, α, 0.9999–0.9840; particle diameter, D_p, $0.3\times 10^{-4} - 5.0\times 10^{-4}$ m; density difference between particle and gas, $\rho_p - \rho_G$, 1228–2700 kg/m^3; pipe diameter, D_T, $6.78\times 10^{-3} - 1.8\times 10^{-2}$ m. To widen the validity limits, the model has to be checked with the other experimental data, especially for large particle sizes and large pipe diameters.

Table 1. Calculated and experimental drag velocities, dispersed solids densities and the calculated external force (Experimental data of Hariu & Molstad, 1949).

System=Air-Ottawa sand ; R_p(m)=2.51×10^{-4}
ρ_p=2643. kg/m^3 ; R_T(m)=3.38×10^{-3}
ρ_G=1.184 kg/m^3 ; L(m) =0.814

SLR	u_G(m/s)	α	$f_{e,s}$(dynes)	u_r^*(m/s)	u_r^{**}	ρ_{ds}^*(kg/m^3)	ρ_{ds}^{**}
6.8	6.68	0.989	7.81×10^{-2}	4.79	4.97	29.31	30.11
12.8	6.68	0.980	7.33×10^{-2}	4.48	4.88	46.93	52.54
18.7	6.68	0.972	7.28×10^{-2}	4.36	4.88	64.39	74.48
30.5	6.68	0.956	7.57×10^{-2}	4.18	4.91	98.99	116.93
5.3	10.76	0.992	2.78×10^{-1}	7.41	7.53	20.34	20.34
8.5	10.79	0.988	2.78×10^{-1}	6.77	7.53	27.71	32.04
11.8	10.82	0.984	2.79×10^{-1}	6.40	7.56	34.92	43.89
15.0	10.85	0.979	2.83×10^{-1}	6.19	7.62	42.13	55.26

System=Air-Ottawa sand ; R_p(m)=2.51×10^{-4}
ρ_p=2643. kg/m^3 ; R_T(m)=6.77×10^{-3}
ρ_G=1.184 kg/m^3 ; L(m)=1.018

SLR	u_G(m/s)	α	$f_{e,s}$(dynes)	u_r^*(m/s)	u_r^{**}	ρ_{ds}^*(kg/m^3)	ρ_{ds}^{**}
2.7	5.21	0.995	1.23×10^{-2}	3.93	3.99	13.14	13.14
4.3	5.21	0.992	1.02×10^{-2}	3.84	3.93	19.54	20.34
6.0	5.18	0.990	7.95×10^{-3}	3.75	3.93	25.95	27.87
7.6	5.18	0.987	7.60×10^{-3}	3.72	3.90	32.36	34.92
1.1	8.99	0.999	1.39×10^{-1}	5.73	5.76	3.68	3.84
2.6	8.99	0.997	1.29×10^{-1}	5.55	5.67	8.00	8.17
4.0	8.96	0.995	1.27×10^{-1}	5.49	5.61	12.33	12.65
5.5	8.93	0.994	1.25×10^{-1}	5.39	5.61	16.50	17.14
1.4	12.34	0.998	3.10×10^{-1}	7.16	7.92	4.00	4.65
2.3	12.31	0.997	3.00×10^{-1}	7.22	7.77	6.73	7.53
3.3	12.28	0.996	2.95×10^{-1}	7.22	7.74	9.45	10.41

System=Air-Sea sand ; R_p(m)=1.07×10^{-4}
ρ_p=2707. kg/m^3 ; R_T(m)=3.38×10^{-3}
ρ_G=1.185 kg/m^3 ; L(m)=0.814

SLR	u_G(m/s)	α	$f_{e,s}$(dynes)	u_r^*(m/s)	u_r^{**}	ρ_{ds}^*(kg/m^3)	ρ_{ds}^{**}
3.9	8.75	0.997	4.13×10^{-2}	3.99	3.93	8.49	8.33
7.4	8.75	0.994	4.16×10^{-2}	3.87	3.96	16.02	15.86
11.0	8.75	0.991	4.18×10^{-2}	3.84	3.99	23.55	23.39
14.6	8.75	0.989	4.19×10^{-2}	3.78	3.96	30.91	30.91

* Experimental values ; ** Calculated (theoretical) values

Another conclusion drawn from table 1 is that for the same gas velocity and the same particle size, the average drag velocity and the external force on a solid particle are independent of SLR. The average drag velocity and the external force increase with the increasing gas superficial velocity and the particle size. This has been supported by Özbelge (1983) with the further calculations using the experimental data of Vogt & White (1948) and Depew (1960).

REFERENCES

BIRD, R. B., STEWART, W. E. & LIGHTFOOT, E. N. 1960 *Transport Phenomena*. Wiley, New York.

BOOTHROYD, R. G. 1971 *Flowing Gas-Solids Suspensions*. Chapman & Hall, London.

DEPEW, C. A. 1960 Heat transfer to flowing gas-solids mixtures in a vertical circular duct. Ph.D. Thesis, University of California, Berkeley, U.S.A.

HARIU, O. H. & MOLSTAD, M. C. 1949 Pressure drop in vertical tubes in transport of solids by gases. *Ind. Engng Chem.* **41**, 1148–1160.

Kramer, T. J. & Depew, C. A. 1972 Analysis of mean flow characteristics of gas-solids suspensions. *J. Basic Engng Trans. ASME* **94**, 731–738.

Metha, N. C., Smith, J. M. & Comings, E. W. 1957 Pressure drop in air-solid flow systems. *Ind. Engng Chem.* **49**, 986–992.

Oki, K., Walawender, W. P. & Fan, L. T. 1977 The measurement of local velocity in solid particles. *Powder Technol.* **18**, 171–178.

ÖZBELGE, T. A. 1983 Ph.D. Thesis, Middle East Technical University, Chem. Eng. Dept., Ankara, Turkey.

ÖZBELGE, T. A. & SOMER, T. G. 1982 Heat transfer to gas-solid suspensions flowing turbulently in a vertical pipe. *Proc. 16th Southeastern Seminar on Thermal Sciences*. Ann Arbor, Michigan, U.S.A.

RIETHMULLER, M. L. 1973 Optical measurement of velocity in particulate flows, *Measurements of Velocities in Single and Two-Phase Flows*, von Karman Inst. for Fluid Dyn. Lecture Series 54.

RIETHMULLER, M. L. 1981 Personal Communication at von Karman Inst. for Fluid Dynamics, Rhode-St-Génesé, Belgium (9 March–6 April).

VOGT, E. G. & WHITE, R. R. 1948 Friction in the flow of suspensions (Granular solids in gases through pipe). *Ind. Engng Chem.* **40**, 1731–1738.

WEN, C. Y. & GALLI, A. F. 1971 *Fluidization* (Edited by DAVIDSON, J. F. & HARRISON, D.). Academic Press, London.